# Numerical Solutions of the Equations in One Variable f(x)=0

## Exercise

*Find the roots of the equation:* **sin x** $= \frac{-3}{x+5}$ *on interval* ⟨-3; 5,5⟩

```
Clear[f]
f[x_] = 3 / (x + 5) + Sin[x]
```

$$\frac{3}{5+x} + \text{Sin}[x]$$

```
Solve[f[x] == 0, x]
```

```
Solve::tdep : The equations appear to involve the
   variables to be solved for in an essentially non-algebraic way. >>
```

$$\text{Solve}\left[\frac{3}{5+x} + \text{Sin}[x] == 0,\ x\right]$$

*To find the roots of the equation numerical methods are to be used.*
*Calculation via numerical methods requires 3 steps:*
*1. Separation of roots.*

*Determination of intervals ⟨a, b⟩; in each interval there exists just one root, the function f is continous, strictly monotone (strictly increasing or decreasing) without any stationary point (local maximum, minimum or inflection points where f'(x)=0 or f''(x)=0), i.e. For all* $x \in \langle a,b \rangle$: $f(a)\, f(b) < 0$, $f'(x) \neq 0$, $f''(x) \neq 0$

*2. Numerical method application.*

*Comparison of Secant method and Newton's method.*

*3. Estimation of the absolute error.*

**Problem:**

*Find the roots of the equation:* **sin x** $= \frac{-3}{x+5}$ *on interval* ⟨-3; 5,5⟩. Use Secant and Newton's methods. Count within the tolerance tol $= 10^{-8}$, i.e. $|f(x_n)| <$ tol.

*Graph of the function within the interval* **⟨-3; 5,5⟩***:*

```
Plot[f[x], {x, -3, 5.5}]
```




*There exist 3 roots of the function within the interval* **⟨-3; 5,5⟩**:

## ■ $1^{st}$ root

### *Separation of the root.*

*Determination of the interval ⟨a,b⟩; where just one root exists, the function f is continous, strictly monotone (strictly decreasing) without any stationary point (local maximum,minimum or inflection points where f'(x)=0 or f''(x)=0).*

```
a = -2; b = -1.5;
Plot[f[x], {x, a, b}]
```




## *Numerical Method Application.*

**SECANT METHOD** (Všeobecná metóda tetív)

**Method of linear interpolation.** $x_0 = a,\ x_1 = b,\ x_{i+1} = x_i - f(x_i)\ \frac{x_{i-1}-x_i}{f(x_{i-1})-f(x_i)}$

### *Counting with no tolerance:*

*The body of the loop involves two items :*

*1. Calculation*

*2. Print out*

```
x[0] = a; x[1] = b;
Do[
 x[i + 1] = x[i] - f[x[i]] * (x[i - 1] - x[i]) / (f[x[i - 1]] - f[x[i]]) // N;
 Print["x(", i + 1, ")=", x[i + 1]],
 {i, 1, 10}]
```

```
x(2)=-1.80372

x(3)=-1.90235

x(4)=-1.8629

x(5)=-1.86508

x(6)=-1.86516

x(7)=-1.86516

x(8)=-1.86516

x(9)=-1.86516

x(10)=-1.86516

Power::infy : Infinite expression 1/0. encountered. >>

∞::indet : Indeterminate expression 0. ComplexInfinity encountered. >>

x(11)=Indeterminate
```

***Notes:***

*x(6) - x(10) seem to be same. To see all machine digits of x(i) just copy the output:*

"x(", 8, ")=" -1.8651605268487734`
"x(", 9, ")=" -1.8651605268487736`
"x(", 10, ")=" -1.8651605268487736`

*The same can be seen when the FullForm statement is used:*

```
FullForm[x[8]]
```

```
-1.8651605268487734`
```

*or*
*the print format of numbers can be set by the statements* ***NumberForm or PaddedForm (see help - F1)***

```
NumberForm[x[8], 16]
```

```
-1.865160526848773
```

*The same loop provided by the NumberForm print out formatting:*

```
x[0] = a; x[1] = b;
Do[
 x[i + 1] = x[i] - f[x[i]] (x[i - 1] - x[i])/(f[x[i - 1]] - f[x[i]]) // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 16]],
 {i, 1, 10}]
```

```
x(2)=-1.80372056456821
x(3)=-1.902347325216691
x(4)=-1.86289615178002
x(5)=-1.865081830045959
x(6)=-1.865160699605912
x(7)=-1.865160526835623
x(8)=-1.865160526848773
x(9)=-1.865160526848774
x(10)=-1.865160526848774
```

```
Power::infy : Infinite expression 1/0. encountered. >>
∞::indet : Indeterminate expression 0. ComplexInfinity encountered. >>
```

```
x(11)=Indeterminate
```

***Task:* Explain the reason, why x (11) is indeterminated**

*Exploring the function values, we can see that* $x_9$= -1.865160526848774 *is the root of the equation*

```
f[x[8]]
```

$-1.11022\times 10^{-16}$

```
f[x[9]]
```

```
0.
```

***Note:*** *The statement* **`FindRoot[f[x],{x,a,b}]`** *works under the similar algorithm.*

```
FindRoot[f[x], {x, a, b}];
NumberForm[%, 16]
```

```
{x → -1.865160526848774}
```

*In case, reaching of roots requires too many iterations, the loop is usually terminated by a break condition*

*Loop with break condition:* $|f(x_i)| < \text{tol}$ (*tolerance*) *and additional output information:*

*The body of the loop involves 3 items :*

1. *Calculation*
2. *Print out*
3. *Break condition*

*After breaking, the information about the distance* $|f(x_n) - 0|$ *and the number of iterations is printed.*

```
Clear[x]
x[0] = a; x[1] = b; tol = 10^(-8);

Do[
 x[i + 1] = x[i] - f[x[i]] * (x[i - 1] - x[i]) / (f[x[i - 1]] - f[x[i]]) // N;
 Print[i, ".iteration: x(", i + 1, ")=", NumberForm[x[i + 1], 10]];

 If[Abs[f[x[i + 1]]] < tol,
              Print["|f(xn)-0|= ", Abs[f[x[i + 1]]]];
              Print["Number of iterations: ", i];
              Break[]],
 {i, 1, 10}]
```

```
1.iteration: x(2)=-1.803720565

2.iteration: x(3)=-1.902347325

3.iteration: x(4)=-1.862896152

4.iteration: x(5)=-1.86508183

5.iteration: x(6)=-1.8651607

6.iteration: x(7)=-1.865160527
```

$|f(x_n) - 0| = 7.82974 \times 10^{-12}$

```
Number of iterations: 6
```

*Table with i,* $x_i$*, and* $f(x_i)$ *columns:*

```
T = Table[{i, x[i], f[x[i]]}, {i, 1, 7}];
TableForm[T, TableHeadings → {None, {"i    ", "     xi      ", "     f(xi)      "}}]
```

| i | $x_i$ | $f(x_i)$ |
|---|---|---|
| 1 | -1.5 | -0.140352 |
| 2 | -1.80372 | -0.0344043 |
| 3 | -1.90235 | 0.0229367 |
| 4 | -1.8629 | -0.00134527 |
| 5 | -1.86508 | -0.000046853 |
| 6 | -1.86516 | $1.02861 \times 10^{-7}$ |
| 7 | -1.86516 | $-7.82974 \times 10^{-12}$ |

*The function value of the 6th iteration* $x_7$= -1.865160527 *differs from the desired zero value by less than tolerance value* $\text{tol} = 10^{-8}$. *It is considered to be the approximation of the root found by Secant method.*

```
r1S = x[7]
```

```
-1.86516
```

***Task:*** ***Try different values of tolerance.***

***Task:*** ***Try different initial values of*** $x_0$, $x_1$ $\in\langle a,b\rangle$. ***Watch the number of iterations.***

***Task:*** ***Try different intervals ⟨a,b⟩. Watch the number of iterations and the value of the root approximation*** $x_n$.

***Be careful about suitable interval specification !***

*If the rules for the interval specification are not kept, the algorithm can lead to the solution far from the interval or may not converge.*

```
Plot[f[x], {x, -2.5, -1.2}]
```




```
Clear[x]
x[0] = -2.5; x[1] = -1.2; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] * (x[i - 1] - x[i]) / (f[x[i - 1]] - f[x[i]]) // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 10]];
 If[Abs[f[x[i + 1]]] < tol,
              Print["|f(xn)-0|= ", Abs[f[x[i + 1]]]];
              Print["Number of iterations: ", i];
              Break[]],
 {i, 1, 30}]
```

```
x(2)=-1.449074992

x(3)=5.647952676

x(4)=-7.848011874

x(5)=8.06257696

x(6)=2.169353499

x(7)=202.5120096

x(8)=1052.41121

....

x(26)=1162.386712
```

$|f(x_n)-0| = 5.01505\times 10^{-13}$

```
Number of iterations: 25
```

```
Plot[f[x], {x, 1160, 1164}]
```




**Estimation of the absolute error** $\text{eps} = \frac{|f(x_n)|}{m}$, $m = \min |f'(x)|$ on $\langle a, b\rangle$

for the root approximation $x_7$ = x1S = -1.865160526835623 on interval <a, b>, a= -2, b=-1.5

```
NumberForm[r1S, 16]
```

```
-1.865160526835623
```

## Minimum value of $|f'(x)|$ on $\langle a, b\rangle$

*Graph of* $|f'(x)|$ *within* $\langle a, b\rangle$

```
a = -2; b = -1.5;
Plot[Abs[f'[x]], {x, a, b}]
```




*$|f'(x)|$ acquires minimum at x =-1.5*

```
m = Abs[f'[-1.5]]
```

```
0.174161
```

## Calculation of the estimation $\text{eps} = \frac{|f(x_n)|}{m}$,

```
eps1S = Abs[f[r1S]] / m
```

$4.4957 \times 10^{-11}$

***Task:*** *Verify that the root* ${}^1x \in \langle r1S - eps1S,\ r1S + eps1S\rangle$, ${}^1x = -1.865160526848774$

```
-1.865160526848774 < r1S + eps1S
```

```
True
```

```
-1.865160526848774 > r1S - eps1S
```

```
True
```

***RESULT:*** ***r1S* = -1.865160526835623** *is the approximation of the first root* ${}^{1}x$ on the interval $\langle -3, 5.5\rangle$, *found by the Secant method with tolerance tol* = $10^{-8}$, ${}^{1}x \in \left\langle r1S - 4.4957\times 10^{-11},\ r1S + 4.4957\times 10^{-11}\right\rangle$

## NEWTON'S METHOD (metóda dotyčníc) $x_0 = a,\ \ x_{i+1} = x_i - \dfrac{f(x_i)}{f'(x_i)}$

1st **root**

```
Plot[f[x], {x, a, b}]
```




```
Clear[x]; a = -2
x[0] = a; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 16]];
 If[Abs[f[x[i + 1]]] < tol,
               Print["|f(xn)-0|=  ", Abs[f[x[i + 1]]]];
              Print["Number of iterations: ", i + 1];
               Break[]],
 {i, 0, 10}]
```

```
-2
```

```
x(1)=-1.878979355532804
x(2)=-1.86534026253019
x(3)=-1.865160558082779
x(4)=-1.865160526848775
```

$|f(x_n)-0| = \ \ 5.55112\times 10^{-16}$

```
Number of iterations: 4
```

*The function value of the 4th iteration* $x_4$= -1.865160526848775 *differs from the desired zero value by less than tolerance* tol = $10^{-8}$. *It is considered to be the approximation of the root found by means of Newton's method within the tolerance* tol = $10^{-8}$.

```
r1N = x[4]
```

```
-1.86516
```

*The FindRoot statement* **FindRoot [f[x], {x, a}]** *works with arguments of Newton's method under the similar algorithm.*

```
r = FindRoot[f[x], {x, a}];
NumberForm[r, 16]
```

{x → -1.865160526848775}

***Estimation of the absolute error*** **eps =** $\frac{|f(x_n)|}{m}$, ***m*** **= min** $|f'(x)|$ on ⟨a, b⟩

for the root approximation $x_4$ = r1N = -1.865160526848775 on interval <a, b>, a= -2, b=-1.5

*m was calculated above*

```
m
```

0.174161

*Calculation of* the estimation eps = $\frac{|f(x_n)|}{m}$

```
eps1N = Abs[f[r1N]] / m
```

$3.18735 \times 10^{-15}$

***RESULT:*** ***r1N*** = -1.865160526848775 *is the approximation of the first root* $^1x$, on the interval ⟨−3, 5.5⟩, *found by the* ***Newton's method*** *with tolerance tol* = $10^{-8}$, $^1x \in \langle r1N - 3.18735 \times 10^{-15},\ r1N + 3.18735 \times 10^{-15} \rangle$

***Task:*** ***Try different initial values of*** $x_0$ ***∈⟨a,b⟩. Watch the number of iterations.***

***Task:*** ***Compare the approximations of the 1st root: r1S (Secant method) and r1N (Newton's method)***
*Pay attention to the number of iterations and the error estimation. Which method is more advantageous?*

***Task:*** ***Build the loop with break condition "eps < ..." (for both methods)***

## ■ 2nd root

***Separation of the root.***

```
Plot[f[x], {x, -1, 0}]
```




***Task:*** *Show that the interval ⟨a,b⟩ satisfies the requirements of convergency.*

***Numerical Method Application.***

**SECANT METHOD** $x_0 = a,\ x_1 = b,\ x_{i+1} = x_i - f(x_i)\,\frac{x_{i-1} - x_i}{f(x_{i-1}) - f(x_i)}$

```
a = -1; b = 0;
Clear[x]
```

```
x[0] = a; x[1] = b; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] * (x[i - 1] - x[i]) / (f[x[i - 1]] - f[x[i]]) // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 16]];
 If[Abs[f[x[i + 1]]] < tol,
              Print["|f(xn)-0|=  ", Abs[f[x[i + 1]]]]];
              Print["Number of iterations: ", i];
              Break[]],
 {i, 1, 10}]
```

```
x(2)=-0.867715367936503

x(3)=-0.81749000340988

x(4)=-0.7927712897461827

x(5)=-0.7942207305785502

x(6)=-0.7941953800103362

x(7)=-0.7941953530456195
```

$|f(x_n)-0|= \ 2.71672\times 10^{-13}$

```
Number of iterations: 6
```

```
r = FindRoot[f[x], {x, a, b}];
NumberForm[r, 16]
```

{x → -0.7941953530461311}

```
r2S = x[7]
```

-0.794195

## *Estimation of the absolute error* eps = $\frac{|f(x_n)|}{m}$, $m = \min |f'(x)|$ on ⟨a, b⟩

for the root approximation $x_n$ = x2S = -0.7941953530456195 on interval <a, b>, a= -1, b=0

## *Minimum value of* $|f'(x)|$ on ⟨a, b⟩

### *Graph of* $|f'(x)|$ *within* ⟨a, b⟩

```
a = -1; b = 0;
Plot[Abs[f'[x]], {x, a, b}]
```




*$|f'(x)|$ acquires minimum at x =-1*

```
m = Abs[f'[-1]] // N
```

```
0.352802
```

*Calculation of the estimation* eps = $\frac{|f(x_n)|}{m}$,

```
eps2S = Abs[f[r2S]] / m
```

$7.70039\times10^{-13}$

***RESULT:*** ***r2S* = -0.7941953530456195** *is the approximation of the second root* $^2x$, on the interval $\langle -3, 5.5\rangle$, *found by the Secant method with tolerance tol* = $10^{-8}$,
$^2x \in \langle r2S - 7.70039\times10^{-13},\ r2S + 7.70039\times10^{-13}\rangle$

## NEWTON'S METHOD (metóda dotyčníc) $x_0 = a,\ x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$

```
a = -1; b = 0;
```

```
Clear[x]
x[0] = a; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 16]];
 If[Abs[f[x[i + 1]]] < tol,
               Print["|f(xn)-0|=  ", Abs[f[x[i + 1]]]]];
              Print["Number of iterations: ", i + 1];
               Break[]],
 {i, 0, 10}]
```

```
x(1)=-0.7407301956748437

x(2)=-0.7922822411638744

x(3)=-0.7941926292906036

x(4)=-0.7941953530405873
```

$|f(x_n)-0| = 2.94509\times10^{-12}$

```
Number of iterations: 4
```

```
r2N = x[4]
```

```
-0.794195
```

```
r = FindRoot[f[x], {x, a}];
NumberForm[r, 16]
```

```
{x → -0.7941953530461308}
```

***Estimation of the absolute error*** **eps =** $\frac{|f(x_n)|}{m}$**, *m* = min |*f* '(*x*)|** on ⟨a, b⟩

for the root approximation $x_n$ = x2N = -0.7941953530405873 on interval <a, b>, a= -1, b=0

*m was calculated above*

```
m
```

```
0.352802
```

*Calculation of* the estimation eps = $\frac{|f(x_n)|}{m}$,

```
eps2N = Abs[f[r2N]] / m
```

$8.3477 \times 10^{-12}$

**RESULT:** ***r2N*** = -0.7941953530405873 *is the approximation of the second root* $^2x$, on the interval $\langle -3, 5.5\rangle$, *found by the* ***Newton's method*** *with tolerance tol* = $10^{-8}$, $^2x \in \left\langle r2N - 8.3477\times 10^{-12},\ r2N + 8.3477\times 10^{-12}\right\rangle$

***Task:*** ***Compare the approximations of the 2nd root: r2S (Secant method) and r2N (Newton's method)***
*Pay attention to the number of iterations and the error estimation. Which method is more advantageous?*

## ■ 3rd root

***Separation of the root.***

```
Plot[f[x], {x, 2, 4}]
```




***Task:*** *Show whether the interval ⟨2,4⟩ satisfies the requirements of convergency.*

***Numerical Method Application.***

**SECANT METHOD** $x_0 = a,\ x_1 = b, x_{i+1} = x_i - f(x_i)\,\frac{x_{i-1}-x_i}{f(x_{i-1})-f(x_i)}$

```
a = 2; b = 4;
Clear[x]
x[0] = a; x[1] = b; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] * (x[i - 1] - x[i]) / (f[x[i - 1]] - f[x[i]]) // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 16]];
 If[Abs[f[x[i + 1]]] < tol,
              Print["|f(xn)-0|= ", Abs[f[x[i + 1]]]];
              Print["delta x= ", Abs[x[i + 1] - x[i]]];
              Print["Number of iterations: ", i];
              Break[]],
 {i, 1, 10}]
```

x(2)=3.519150602782648

x(3)=3.499650637703172

x(4)=3.502215597180993

x(5)=3.502207442371615

$|f(x_n)-0|= 3.77202 \times 10^{-9}$

delta x= $8.15481 \times 10^{-6}$

Number of iterations: 4

```
r = FindRoot[f[x], {x, a, b}];
NumberForm[r, 16]
```

{x → 3.502207438511514}

```
r3S = x[5]
```

3.50221

***Estimation of the absolute error*** **eps =** $\frac{|f(x_n)|}{m}$**, *m* = min |*f* '(*x*)|** on ⟨a, b⟩

for the root approximation $x_5$ = r3S =3.502207442371615 on interval <a, b>, a= 2, b=4

```
a = 2; b = 4;
Plot[Abs[f'[x]], {x, a, b}]
```




*|f '(x)| acquires minimum at x =2*

```
m = Abs[f'[2]] // N
```

0.477371

```
f[r3S]
```

$-3.77202\times10^{-9}$

*Calculation of* the estimation eps = $\frac{|f(x_n)|}{m}$,

```
eps3S = Abs[f[r3S]] / m
```

$7.90164\times10^{-9}$

***RESULT:*** ***r3S*** = 3.502207442371615 *is the approximation of the third root* $\sqrt[3]{x}$, on the interval ⟨−3, 5.5⟩, *found by the* ***Secant method*** *with tolerance tol* = $10^{-8}$, $\sqrt[3]{x} \in \langle r3S - 7.90164\times10^{-9},\ r3S + 7.90164\times10^{-9}\rangle$

$$x_0 = a,\quad x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$$

## NEWTON'S METHOD (metóda dotyčníc) $x_0 = a,\ x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$

```
Clear[x]
x[0] = a; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 16]];
```

```
If[Abs[f[x[i + 1]]] < tol,
             Print["|f(xn)-0|=  ", Abs[f[x[i + 1]]]];
            Print["Number of iterations: ", i + 1];
             Break[]],
   {i, 0, 10}]
```

x(1)=4.802574812454607

x(2)=16.52688149782068

x(3)=15.66965284819256

x(4)=15.85194791728253

x(5)=15.85233303619421

x(6)=15.85233304693113

$|f(x_n)-0|=$  $6.93889\times10^{-16}$

Number of iterations: 6

***Be careful about suitable interval specification !*** $x_6 \notin \langle a,\ b\rangle$

```
Plot[f[x], {x, 2, 4}]
```




```
Plot[f'[x], {x, 2, 4}]
```




```
Plot[f''[x], {x, 2, 4}]
```




```
FindRoot[f''[x] == 0, {x, a}]
```

{x → 3.13043}

***<u>Task:</u> Explain why the above interval specification is not appropriate for the Newton's method***

```
a = 3.3; b = 3.7;
Plot[f[x], {x, a, b}]
```




```
Clear[x]
x[0] = a; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 16]];
 If[Abs[f[x[i + 1]]] < tol,
               Print["|f(xn)-0|=  ", Abs[f[x[i + 1]]]];
              Print["Number of iterations: ", i + 1];
               Break[]],
 {i, 0, 10}]
```

x(1)=3.497569994936402

x(2)=3.502203486809872

x(3)=3.502207438508616

$|f(x_n)-0|=\quad 2.83135\times 10^{-12}$

Number of iterations: 3

```
r3N = x[3]
```

3.50221

```
r = FindRoot[f[x], {x, a}];
NumberForm[r, 16]
```

```
{x → 3.502207438511514}
```

***Estimation of the absolute error*** **eps =** $\frac{|f(x_n)|}{m}$**,** $m = \min |f'(x)|$ on ⟨a, b⟩

for the root approximation $x_3$ = x3N = 3.502207438508616 on interval <a, b>, a= 3.3, b=3.7

```
a = 3.3; b = 3.7;
Plot[Abs[f'[x]], {x, a, b}]
```




*|f '(x)| acquires minimum at x =*3.7

```
m = Abs[f'[b]] // N
```

```
0.887735
```

*Calculation of* the estimation eps = $\frac{|f(x_n)|}{m}$,

```
eps3N = Abs[f[r3N]] / m
```

$3.1894 \times 10^{-12}$

***RESULT:*** ***r3N*** **= 3.502207438508616** *is the approximation of the third root* $^3x$,
on the interval ⟨−3, 5.5⟩, *found by the* ***Newton's method*** *with tolerance tol* = $10^{-8}$,
$^3x \in \langle r3N - 3.1894 \times 10^{-12},\ r3N + 3.1894 \times 10^{-12} \rangle$

**Dangerous situations:**

**Be careful about stationary points (extrema, infl. point)** $f'(x_0)$ **and points where** $f''(x_0) = 0$ **!**

**Start point in vicinity of extremum:**

```
Plot[f[x], {x, -2, 2}]
```




```
FindRoot[f'[x] == 0, {x, -1.5}]
```

```
{x → -1.34438}
```

```
Clear[x]
x[0] = -1.344; tol = 10^(-8);
Do[
 x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 Print["x(", i + 1, ")=", NumberForm[x[i + 1], 10]];
 If[Abs[f[x[i + 1]]] < tol,
               Print["|f(xn)-0|=  ", Abs[f[x[i + 1]]]];
              Print["Number of iterations: ", i + 1];
               Break[]],
 {i, 0, 10}]
```

x(1)=371.4740526

x(2)=370.5007985

x(3)=370.7027878

x(4)=370.699948

x(5)=370.6999479

$|f(x_n)-0| = \ 3.33067 \times 10^{-15}$

Number of iterations: 5

```
Plot[f[x], {x, 362, 373}]
```




```
FindRoot[f[x] == 0, {x, -1.344}]
```

{x → 22.1021}

**Work out individualy:**

*Find the roots of the equation:* **ln *x* - sin *x* =*x* – 2** *on interval* ⟨0; 5⟩.

Problems:
1. How many roots does the equation have on the interval ⟨0; 5⟩?
2. Separate the first (second, ...) root. Show that your separation is proper.
3. Find its value $x_n$ via the method of linear interpolation (Secant method) within the tolerance tol, where $|f(x_n)| < tol$. Calculate the estimation of the absolute error $\varepsilon$. Compare it with the result found through the FindRoot command with arguments of this method.
4. Find its value $x_n$ via the Newton's method within the tolerance tol, where $|f(x_n)| < tol$. Calculate the estimation of the absolute error $\varepsilon$. Compare it with the result found through the FindRoot command with arguments of this method.
5. Compare the methods. Which one is more advantageous?
6. Put down at least one point, which is not suitable as a start point in Newton's method. Give the reason.
7. Put down at least one couple of points, which are not suitable as start points in Secant method. Give the reason.
8. Figure out the value of the 3.rd iteration by hand.
9. Find the approximations satisfying prescribed accuracy $\varepsilon < 10^{-9}$